Lesson no.37

Device Drivers

Device drivers are operating system extensions that become part of the operating system and extend its services to new devices. Device drivers in DOS are very simple. They just have their services exposed through the file system interface. 

Device driver file starts with a header containing a link to the next driver in the first four bytes followed by a device attribute word. The most important bit in the device attribute word is bit 15 which dictates if it is a character device or a block device. If the bit is zero the device is a character device and otherwise a block device. Next word in the header is the offset of a strategy routine, and then is the offset of the interrupt routine and then in one byte, the number of units supported is stored. This information is padded with seven zeroes.

Strategy routine is called whenever the device is needed and it is passed a request header. Request header stores the unit requested, the command code, space for return value and buffer pointers etc. Important command codes include 0 to initialize, 1 to check media, 2 to build a BIOS parameter block, 4 and 8 for read and write respectively. For every command the first 13 bytes of request header are same. 

RH+0
BYTE

Length of request header

RH+1
BYTE

Unit requested

RH+2
BYTE

Command code

RH+3
BYTE

Driver’s return code

RH+5
9 BYTES
Reserved

The request header details for different commands is listed below.

0 – Driver Initialization

Passed to driver

RH+18
DWORD
Pointer to character after equal sign on CONFIG.SYS line that loaded driver (read-only)

RH+22
BYTE
Drive number for first unit of this block driver (0=A...) 

Return from driver

RH+13
BYTE
Number of units (block devices only)

RH+14
DWORD 
Address of first free memory above driver (break address)

RH+18
DWORD
BPB pointer array (block devices only)

1 – Media Check

RH+13
BYTE
Media descriptor byte

Return

RH+14
BYTE
Media change code



-1 if disk changed



0 if dont know whether disk changed



1 if disk not changed

RH+15
DWORD
pointer to previous volume label if device attrib bit 11=1 (open/close/removable media supported)

2 – Build BPB

RH+13
BYTE
Media descriptor byte

RH+14
DWORD 
buffer address (one sector)

Return 

RH+18
DWORD
pointer to new BPB

if bit 13 (ibm format) is set buffer is first sector of fat, otherwise scrach space

4 – Read / 8 – Write / 9 – Write with verify

RH+13
BYTE 
Media descriptor byte

RH+14
DWORD 
transfer address

RH+18
WORD
byte or sector count

RH+20
WORD
starting sector number (for block devices)

Return
RH+18
WORD
actual byte or sectors transferred

RH+22
DWORD
pointer to volume label if error 0Fh is returned

The BIOS parameter block discussed above is a structure that provides parameters about the storage medium. It is stored in the first sector or the boot sector of the device. Its contents are listed below. 

00-01
bytes per sector

02
sectors per allocation unit

03-04
Number of reserved sectors ( 0 based)

05
number of file allocation tables

06-07
max number of root directory entries

08-09
total number of sectors in medium

0A
media descriptor byte

0B-0C
number of sectors occupied by a single FAT

0D-0E
sectors per track (3.0 or later)

0F-10
number of heads (3.0 or later)

11-12
number of hidden sectors (3.0 or later)

13-14
high-order word of number of hidden sectors (4.0)

15-18
IF bytes 8-9 are zero, total number of sectors in medium

19-1E
Reserved should be zero

We will be building an example device driver that takes some RAM and expresses it as a secondary storage device to the operating system. Therefore a new drive is added and that can be browsed to, filed copied to and from just like ordinary drives expect that this drive is very fast as it is located in the RAM. This program cannot be directly executed since it is not a user program. This must be loaded by adding the line “device=filename.sys” in the “config.sys” file in the root directory. 

	; ram disk dos block device driver
header:     dd   -1                  ; no next driver
            dw   0x2000              ; driver attributes: block device
            dw   strategy            ; offset of strategy routine
            dw   interrupt           ; offset of interrupt routine
            db   1                   ; no of units supported
            times 7 db 0             ; reserved
request:    dd   0                   ; space for request header
ramdisk:    times 11 db 0            ; initial part of boot sector

bpb:        dw   512                 ; bytes per sector

            db   1                   ; sectors per cluster

            dw   1                   ; reserved sectors

            db   1                   ; fat copies

            dw   48                  ; root dir entries

            dw   105                 ; total sectors

            db   0xf8                ; media desc byte: fixed disk

            dw   1                   ; sectors per fat

            times 482 db 0           ; remaining part of boot sector

            db   0xfe, 0xff, 0xff    ; special bytes at start of FAT

            times 509 db 0           ; remaining FAT entries unused

            times 103*512 db 0       ; 103 sectors for data
bpbptr:     dw   bpb                 ; array of bpb pointers

dispatch:   dw   init                ; command 0: init
            dw   mediacheck          ; command 1: media check
            dw   getbpb              ; command 2: get bpb
            dw   unknown             ; command 3: not handled
            dw   input               ; command 4: input
            dw   unknown             ; command 5: not handled
            dw   unknown             ; command 6: not handled
            dw   unknown             ; command 7: not handled
            dw   output              ; command 8: output
            dw   output              ; command 9: output with verify
; device driver strategy routine
strategy:   mov  [cs:request], bx    ; save request header offset
            mov  [cs:request+2], es  ; save request header segment
            retf

; device driver interrupt routine
interrupt:  push ax

            push bx

            push cx

            push dx

            push si

            push di

            push ds

            push es

            push cs

            pop  ds

            les  di, [request]

            mov  word [es:di+3], 0x0100

            mov  bl, [es:di+2]

            mov  bh, 0

            cmp  bx, 9

            ja   skip

            shl  bx, 1

            call [dispatch+bx]

skip:       pop  es

            pop  ds

            pop  di

            pop  si

            pop  dx

            pop  cx

            pop  bx

            pop  ax

            retf

mediacheck: mov  byte [es:di+14], 1

            ret

getbpb:     mov  word [es:di+18], bpb

            mov  [es:di+20], ds

            ret

input:      mov  ax, 512

            mul  word [es:di+18]

            mov  cx, ax

            mov  ax, 512

            mul  word [es:di+20]

            mov  si, ax

            add  si, ramdisk

            les  di, [es:di+14]

            cld

            rep  movsb

            ret

output:     mov  ax, 512

            mul  word [es:di+18]

            mov  cx, ax

            lds  si, [es:di+14]

            mov  ax, 512

            mul  word [es:di+20]

            mov  di, ax

            add  di, ramdisk

            push cs

            pop  es

            cld

            rep  movsb

unknown:    ret

init:       mov  ah, 9

            mov  dx, message

            int  0x21

            mov  byte [es:di+13], 1

            mov  word [es:di+14], init

            mov  [es:di+16], ds

            mov  word [es:di+18], bpbptr

            mov  [es:di+20], ds

            ret

message:    db   13, 10, 'RAM Disk Driver loaded',13,10,'$'


